



Hochschule für Technik,  
Wirtschaft und Kultur Leipzig



Abschlussarbeit  
zur Erlangung des akademischen Grades

Bachelor of Science (B.Sc.)

im Studiengang Informatik  
der Fakultät Informatik und Medien

# **Erfüllbarkeit und Optimierung von Pressenplanungsproblemen mit Constraint-Systemen**

Julian Bruder

Leipzig, den 16. September 2024

Erstgutachter: Prof. Dr. Johannes Waldmann

Zweitgutachter: Dr. Sebastian Möbius

Die Pressenplanung von Holzbalken aus Bauteilen erfordert eine fein abgestimmte Konfiguration, um die Pressen optimal auszulasten und den Verschnitt der Balken zu minimieren. Dafür hat die Germanedge Solutions GmbH<sup>1</sup>, Industriepartner dieser Arbeit, bereits eine unternehmensinterne Heuristik, welche aufgrund ihrer Natur allerdings möglicherweise (optimale) Lösungen nicht berücksichtigt. Im Rahmen dieser Arbeit wird daher eine prototypische Lösung mithilfe der Methoden der Constraint-Programmierung, speziell jener der Satisfiability Modulo Theories (SMT), erarbeitet, welche optimale Lösungen finden soll. Dazu werden die Constraints der Domäne zunächst mithilfe der Prädikatenlogik modelliert und anschließend in einer konkreten Repräsentation des Modells implementiert.

Dabei wird ein Programm entwickelt, welches aus den textuellen Eingaben der Pressen- und Bauteilspezifikationen ein Constraint-System erzeugt, das anschließend von einem externen Constraint-Solver gelöst wird. Das Programm wandelt die Lösung des Solvers in eine, zu den bereits existenten Schnittstellen passende, textuelle Ausgabe um. Weiter ist das Programm so strukturiert, dass verschiedene Logiken aus SMT zur Erzeugung des Constraint-Systems genutzt werden können, um die performanteste Kodierung des Problems in jenen Logiken einfach zu ermitteln. Dies wird durch Wahl der Programmiersprache Haskell und deren ausdrucksstarkes Typsystem ermöglicht. Als Grundlage zur Kodierung des Pressenplanungsproblems und Interaktion mit externen Solvern wird die Haskell-Bibliothek `Hasmtlib` [Bru24a] verwendet und an geeigneten Stellen erweitert.

Das Ziel für die prototypische Lösung des Problems der Pressenplanung ist die korrekte, optimale und bestenfalls schnelle (Wunsch Germanedge: unter sechs Minuten) Lösung bekannter Testfälle. Da es sich hier um eine prototypische Lösung handelt, werden einige Sonderfälle des Pressenplanungsproblems in dieser Arbeit nicht betrachtet. Am Ende wird die Performance der prototypischen Lösung mit jener der aktuell existenten heuristischen verglichen.

---

<sup>1</sup><https://www.germanedge.com/>, abgerufen am 21.08.2024

# Inhaltsverzeichnis

<b>1. Einleitung</b>	<b>1</b>
<b>2. Problemdomäne Pressenplanung</b>	<b>5</b>
2.1. Anforderungen . . . . .	5
2.2. Modellierung . . . . .	7
2.3. Optimierung . . . . .	8
<b>3. Satisfiability Modulo Theories</b>	<b>10</b>
3.1. Einführung in das Erfüllbarkeitsproblem SMT . . . . .	10
3.2. Sprachstandard SMTLib Version 2.6 . . . . .	10
3.3. SMT-Kodierung von Anzahl-Constraints . . . . .	10
3.4. Optimierung von SMT-Problemen . . . . .	12
<b>4. Haskell und Hasmtlib</b>	<b>13</b>
4.1. Funktionale Programmierung mit Haskell . . . . .	13
4.2. Hasmtlib - Haskell-Bibliothek für SMTLib . . . . .	13
4.2.1. Ergänzung um Anzahl-Constraints . . . . .	16
4.2.2. Ergänzung um Incremental Refinement . . . . .	16
<b>5. Implementierung</b>	<b>17</b>
5.1. Datentypen und Variablen . . . . .	17
5.2. Vorverarbeitung . . . . .	20
5.3. Sorten-polymorphe Kodierung . . . . .	20
5.4. Kodierung der Constraints . . . . .	20
5.5. IO . . . . .	20
5.6. Parallelisierung . . . . .	21
5.7. Modifikation und Konfiguration der Kodierung . . . . .	21
<b>6. Auswertung und Laufzeitmessung</b>	<b>22</b>
6.1. Vergleich mit realistischem Timeout . . . . .	22
6.2. Vergleich mit liberalem Timeout . . . . .	22
<b>7. Zusammenfassung</b>	<b>23</b>
<b>Literatur</b>	<b>24</b>
<b>Appendix</b>	<b>27</b>
A. Beispiel 1 . . . . .	27
B. Beispiel n . . . . .	27

# 1. Einleitung

Die Germanedge Solutions GmbH spezialisiert sich in der Digitalisierung der Produktion und unterstützt Unternehmen mit individuellen Softwarelösungen für die Industrie 4.0 [VAB18]. Darunter auch ein nicht genannter Kunde, welcher Germanedge mit der Optimierung der Planung von Holzpressen beauftragt hat. Dabei werden Holzbauteile zu Holzbalken verleimt und anschließend in Pressen gepresst. Die Art und Weise der Anordnung der Bauteile in den Pressen wird als Pressenplan bezeichnet. Abbildung 1.1 visualisiert einen solchen Pressenplan. Die Presse (grau) presst in Richtung der Pfeile nach unten. Die Balken (gelb) liegen horizontal und bestehen aus Bauteilen, welche durch die vertikalen (schwarzen) Linien getrennt sind. Der Rest jedes Balkens ist rot hervorgehoben.



Abbildung 1.1.: Darstellung eines Pressenplans

An einen Pressenplan werden verschiedene Anforderungen wie beispielsweise die Gleichheit der Längen aller Balken in einer Presse gestellt. Neben diesen Anforderungen sind auch einige Optimierungsziele für den Pressenplan von Bedeutung. Darunter zum Beispiel die Minimierung des Rests, der Balken auf die gleiche Länge auffüllt und somit zusätzliche Materialkosten bedeutet.

Germanedge konnte für dieses Problem bereits eine Lösung finden, diese ist allerdings unter Firmenverschluss. Es ist lediglich bekannt, dass jene einen heuristischen Ansatz wählt. Das ermöglicht zwar einerseits das schnelle Ermitteln von Lösungen nahe am Optimum, andererseits gehen dabei aber auch (optimale) Lösungen verloren.

Daher wird in dieser Arbeit ein neuer, vollständiger Ansatz zur Lösung des Problems gewählt. Dieser besteht in der Kodierung der Anforderungen an den Pressenplan als Constraints in einem Constraint-System. Ein Constraint ist hier eine prädikatenlogische Formel, welche unter einer erfüllenden Belegung den Wahrheitswert **Wahr** haben muss. Ein Constraint-System besteht aus der Konjunktion mehrerer Constraints.

In dieser Arbeit wird das Constraint-System für die Satisfiability Module Theories (kurz SMT) mithilfe des Sprachstandards SMTLib Version 2.6 [BFT17] kodiert. Zu typischen Anwendungsbereichen von SMT gehören neben der Programmverifikation [BT18] auch Ressourcenzuordnungsprobleme [Ans+11; LLC21]. Das Pressenplanungsproblem kann als Ressourcenzuordnungsproblem betrachtet werden, sodass sich die Frage stellt, wie es als SMT-Problem in SMTLib kodiert werden kann. Eine Beispieleingabe von Bauteilen könnte folgende sein:

Länge in mm	Höhe in mm	Anzahl
7000	400	1
13000	400	2
3000	400	3
6000	400	1
9000	400	1
7500	400	1
5000	400	1
5500	400	1
3500	400	1

Tabelle 1.1.: Beispieleingabe von Bauteilen, jede Zeile entpricht einer Bauteilspezifikation

Da das Formulieren der Constraints für einen Pressenplan mit gegebenen Bauteilen händisch in der Constraint-Sprache SMTLib sehr schwer- und fehleranfällig ist (siehe Listing 2), erzeugt ein Programm in einer Gastsprache das Constraint-System für die Constraint-Sprache. Für diese Arbeit wird als jene Gastsprache Haskell [Hud+07] gewählt. Mit der Haskell-Bibliothek `Hasmtlib` [Bru24a] lässt sich beispielhaft das bereits erwähnte Constraint gleicher Länge aller Balken in einer Presse folgend formulieren:

```

1 barLength :: [Bar (Expr RealSort)] -> Expr BoolSort
2 barLength = all
3     (\(b1,b2) -> (b1^.press === b2^.press) ==>
4         (b1^.length === b2^.length)
5     ) . binom

```

Listing 1: Haskell-Code für das Constraint:  $\forall \{b_1, b_2\} \in \binom{Balken}{2} : presse(b_1) = presse(b_2) \rightarrow länge(b_1) = länge(b_2)$

Die Bibliothek erzeugt dann das Constraint-System in der Constraint-Sprache SMTLib. Folgend ein Ausschnitt der erzeugten SMTLib-Kodierung für das Constraint aus Listing 1:

```

...
(declare-fun var_69 () Real)
(declare-fun var_70 () Bool)
(declare-fun var_71 () Real)
...
(assert (and (or var_70 (and (>= var_69 1.44) (<= var_69 2.3))) ...))
(assert (and (or var_70 (and (>= var_68 6.0) (<= var_68 24.15))) ...))
(assert (and (and (and (and (and (and (and (= var_28 (+ (+ ...)))))))))))
...

```

Listing 2: Ausschnitt der Kodierung eines Pressenplanungsproblems

Die Kodierung wird an einen SMT-Solver gegeben, welcher die Erfüllbarkeit des Problems bestimmt und für den Fall *erfüllbar* eine erfüllende Belegung für die Variablen des Problems ermittelt. Zum Beispiel:

```

1  sat
2  (
3    ...
4    (define-fun var_69 () Real    (/ 287 125))
5    (define-fun var_70 () Bool    false)
6    (define-fun var_71 () Real    (/ 483 20))
7    ...
8  )

```

Listing 3: Ausschnitt des Solver-Outputs der Lösung eines Pressenplanungsproblems

Jene Belegung transformiert das Programm dann in das gewünschte Ausgabeformat:

Presse	Schicht	Position	Länge in mm	Höhe in mm
1	1	1	14000	400
1	1	2	7500	400
1	2	1	3000	400
1	2	2	6500	400
1	2	3	3000	400
...	...	...	...	...

Tabelle 1.2.: Beispielausgabe des Pressenplans, Visualisierung in Abbildung 1.1

Die Struktur dieser Arbeit ist folgende:

in Kapitel 2 wird das Problem der Pressenplanung erklärt, modelliert und nach, für das Problem relevanten, Zielfunktionen optimiert.

Folgend wird in Kapitel 3 gezeigt, wie Erfüllbarkeitsprobleme der Satisfiability Modulo Theories im Sprachstandard SMTLib Version 2.6 kodiert werden können.

In Kapitel 4 wird auf einige notwendige Grundlagen von Haskell und die Haskell-Bibliothek `Hasmtlib` eingegangen. Zusätzlich werden dort einige Erweiterungen von `Hasmtlib` diskutiert.

Danach wird die Kodierung des in Abschnitt 2.2 erstellten Modells des Pressenplanungsproblems mithilfe von `Hasmtlib` in Kapitel 5 erläutert.

Weiter wird die Performance der Kodierung in Kapitel 6 mithilfe von Laufzeitmessungen ausgewertet und mit der bereits existenten Heuristik verglichen.

Abschließend wird das Ergebnis dieser Arbeit in Kapitel 7 zusammengefasst.

## 2. Problemdomäne Pressenplanung

In diesem Kapitel wird das Problem der Pressenplanung zunächst erläutert, bevor die prädikatenlogische Modellierung des Problems in einem Modell erfolgt. Abschließend wird das Modell um Optimierungsziele erweitert.

### 2.1. Anforderungen

Im Pressenplanungsproblem sind drei Entitäten und deren Eigenschaften von elementarer Bedeutung: Pressen, Balken und Bauteile. Diese haben jeweils die Eigenschaften Höhe, Breite und Länge. Zusätzlich gibt es Holzlamellen, welche durch Zersägen eines sogenannten *Endlosbretts* entstehen. Dieses kann von einer Maschine *scheinbar* endlos ausgegeben werden. Höhe und Breite der Lamelle sind fest. Lamellen werden vertikal zu Bauteilen verleimt, sodass Bauteile immer die Form eines Quaders haben. Diese Verleimung ist der eigentliche Grund für die Notwendigkeit des Pressens. Folgend eine Abbildung verleimter Lamellen:



Abbildung 2.1.: Vertikale Verleimung von Lamellen

Bauteile wiederum formen Balken - ebenfalls Quader. Bauteile erhält man durch vertikales Zersägen eines Balkens. Folgend eine Abbildung eines Balkens aus drei Bauteilen mit Rest:



Abbildung 2.2.: Balken aus drei Bauteilen mit Rest, Sägelinien schwarz gepunktet

Balken werden vertikal in Pressen gestapelt und anschließend gepresst. Eine mit Balken befüllte Presse, bei der festgelegt ist wie die Balken zu Bauteilen nach der Pressung zersägt werden müssen, nennt man auch Pressenplan. In einem Pressenplan ist jedem Bauteil höchstens eine Position in einer Schicht (von unten gezählter Balken) einer Presse zugeordnet. Folgend die Visualisierung eines Pressenplans für eine Presse:





Abbildung 2.3.: Pressenplan für eine Presse mit vier Balken, 13 Bauteilen und zwei Reststücken

Da Pressen im Allgemeinfeld in einer Pressung nur Bauteile gleicher Breite pressen können, werden die Bauteile nach Breite vorsortiert. Damit reduziert man die Dimension des Problems vom 3-Dimensionalen auf das 2-Dimensionale, da nur noch Länge und Höhe aller Entitäten berücksichtigt werden muss.

Die Eingabe für das Pressenplanungsproblem ist eine Liste von Bauteilen mit Länge und Höhe sowie deren Anzahl. Diese Eingabe wird auch Auftrag genannt. Folgend ein beispielhafter Auftrag:

Länge in mm	Höhe in mm	Anzahl
7000	400	1
13000	400	2
3000	400	3
6000	400	1
9000	400	1
7500	400	1
5000	400	1
5500	400	1
3500	400	1

Tabelle 2.1.: Beispielauftrag, jede Zeile entspricht einer Bauteilspezifikation

Die Lösung des Pressenplanungsproblems (Ausgabe) ist ein Pressenplan. Für den Auftrag in Tabelle 2.1 könnte beispielsweise folgende Ausgabe den Pressenplan beschreiben:

Presse	Schicht	Position	Länge in mm	Höhe in mm
1	1	1	14000	400
1	1	2	7500	400
1	2	1	3000	400
1	2	2	6500	400
1	2	3	3000	400
...	...	...	...	...

Tabelle 2.2.: Beispielausgabe des Pressenplans zu Auftrag in Tabelle 2.1, Visualisierung in Abbildung 2.3

An einen Pressenplan bestehen folgende Anforderungen:

- Pressen haben eine Minimal- und Maximalhöhe. Die Summe der Höhe aller Balken einer Presse muss in diesem Intervall liegen.
- Pressen haben eine Minimal- und Maximallänge. Die Länge aller Balken einer Presse muss in diesem Intervall liegen.
- Alle Balken einer Presse haben die gleiche Länge. Dazu kann der Balken so aus Bauteilen zusammengesetzt sein, dass ein Rest übrig bleibt.
- Alle Bauteile, die einen Balken formen, haben die gleiche Höhe.
- Jedes Bauteil eines Auftrages ist höchstens einem Balken zugeordnet.
- Jeder Balken ist genau einer Presse zugeordnet.

## 2.2. Modellierung

In diesem Abschnitt wird ein mögliches Modell zur Modellierung des Pressenplanungsproblems erläutert. Alle Namen von Funktionen in jenem Modell werden englisch formuliert, um die Nähe zur später in Abschnitt 5.4 gezeigten Implementierung zu wahren. Pressen heißen also *Press*, Balken (Schichten) werden *Layer* genannt und Bauteile erhalten den Namen *Component*. Die Menge aller Pressen wird mit  $P$ , die Menge aller Schichten mit  $L$  und die Menge aller Bauteile mit  $C$  abgekürzt. Weiter werden boolesche als auch arithmetische Operatoren vorausgesetzt und in den Signaturen der Modelle nicht genannt.

Folgend die Signatur  $\Sigma = (\Sigma_F^P \cup \Sigma_F^L \cup \Sigma_F^C, \Sigma_R)$  mit Variablen  $\mathbb{X} = \mathbb{X}^P \cup \mathbb{X}^L \cup \mathbb{X}^C$  wobei

$$\begin{aligned} \Sigma_F^P &= \left\{ \begin{array}{ll} (id, & P \rightarrow \mathbb{N}_0), \\ (minLength, & P \rightarrow \mathbb{N}_0), \\ (maxLength, & P \rightarrow \mathbb{N}_0), \\ (length, & P \rightarrow \mathbb{N}_0), \\ (minHeight, & P \rightarrow \mathbb{N}_0), \\ (maxHeight, & P \rightarrow \mathbb{N}_0), \\ (height, & P \rightarrow \mathbb{N}_0), \\ (layers, & P \rightarrow L_P \subseteq L) \end{array} \right\} & \mathbb{X}^P = \bigcup_{p \in P} \{length(p), height(p)\} \\ \Sigma_F^L &= \left\{ \begin{array}{ll} (id, & L \rightarrow \mathbb{N}_0), \\ (length, & L \rightarrow \mathbb{N}_0), \\ (height, & L \rightarrow \mathbb{N}_0), \\ (waste, & L \rightarrow \mathbb{N}_0), \\ (isEmpty, & L \rightarrow \mathbb{B}) \end{array} \right\} & \mathbb{X}^L = \bigcup_{l \in L} \{length(l), height(l), waste(l), isEmpty(l)\} \\ \Sigma_F^C &= \left\{ \begin{array}{ll} (length, & C \rightarrow \mathbb{N}_0), \\ (height, & C \rightarrow \mathbb{N}_0) \end{array} \right\} & \mathbb{X}^C = \emptyset \\ \Sigma_R &= \{(InLayer, C \times L \rightarrow \mathbb{B})\} & \mathbb{X}^{InLayer} = InLayer \end{aligned}$$

Damit lassen sich folgende Constraints formulieren:

$$\forall p \in P : minHeight(p) \leq height(p) \leq maxHeight(p) \quad (2.1)$$

$$\forall p \in P : minLength(p) \leq length(p) \leq maxLength(p) \quad (2.2)$$

$$\forall p \in P, \forall l \in layers(P) : \neg isEmpty(l) \rightarrow length(p) = length(l) \quad (2.3)$$

$$\forall c \in C, \forall l \in L : InLayer(c, l) \rightarrow height(c) = height(l) \quad (2.4)$$

$$\forall c \in C : atMost(1, \{InLayer(c, l) \mid l \in L\}) \quad (2.5)$$

$$\forall l \in L : isEmpty(l) \leftrightarrow \neg \bigvee_{c \in C} InLayer(c, l) \quad (2.6)$$

$$\forall l \in L : isEmpty(l) \leftrightarrow (length(l) = 0 \wedge height(l) = 0) \quad (2.7)$$

$$\forall p \in P : height(p) = \sum_{l \in layers(P)} height(l) \quad (2.8)$$

$$\forall l \in L : length(l) = waste(l) + \sum_{c \in C} \begin{cases} length(c) & , InLayer(c, l) \\ 0 & , \text{sonst} \end{cases} \quad (2.9)$$

## 2.3. Optimierung

Das obige Modell beschreibt beliebige Lösungen für das Problem der Pressenplanung. Um Kosten- und Zeitaufwand beim Pressen zu reduzieren, sind Pressenpläne

zusätzlich nach folgenden Kriterien lexikografisch zu optimieren (Vorgabe Germa-nedge):

1.  $\min \left( \left| \left\{ c \in C \mid \neg \bigvee_{l \in L} InLayer(c, l) \right\} \right| \right)$
2.  $\min(|P|)$
3.  $\min \left( \sum_{l \in L} waste(l) \right)$

### 3. Satisfiability Modulo Theories

In diesem Kapitel wird zuerst das Erfüllbarkeitsproblem der Satisfiability Module Theories - kurz SMT - vorgestellt. Danach wird auf die Kernelemente des Sprachstandards SMTLib Version 2.6 [BFT17] eingegangen. Dieser ermöglicht die Kodierung von SMT-Problemen für SMT-Solver [BT18]. Ein SMT-Solver - kurz Solver - ist ein Programm, welches für ein kodiertes Problem die Erfüllbarkeit dieses und für den Fall *erfüllbar* (englisch *satisfiable* - kurz *sat*) ein Modell für jenes bestimmen kann. Einige bekannte Solver sind zum Beispiel Z3 [DB08], CVC5 [Bar+22], Yices [Dut14] und OpenSMT [Bru+10]. Diese und viele andere nehmen jährlich an dem internationalen Wettbewerb SMT-COMP [BMS05] für SMT-Solver teil, wo sie sich in verschiedenen Disziplinen miteinander messen. Nach dem Überblick über SMTLib werden Anzahl-Constraints in SMT betrachtet, bevor am Ende dieses Kapitels beleuchtet wird, wie SMT-Probleme nach Zielfunktionen optimiert werden können.

#### 3.1. Einführung in das Erfüllbarkeitsproblem SMT

Logiken, Sorten erklären. Welche Logiken hier relevant? Wie lösen die Solver die Probleme der Logiken? Wie performant lösen die Solver diese - Komplexität? Aktuelle SMT-Comp - beste Solver für relevante Kategorien nennen.

#### 3.2. Sprachstandard SMTLib Version 2.6

Standard Version 2.6: [BFT17]

#### 3.3. SMT-Kodierung von Anzahl-Constraints

Oftmals werden zur Kodierung von SMT-Problemen auch Anzahl-Constraints benötigt [KGK19]. Auf einer Menge von  $n$  gegebenen booleschen Variablen  $\{b_0, b_1, \dots, b_{n-1}\}$  soll dabei die Anzahl wahr belegter Variablen nach  $k$  restriktiert werden. Dabei existieren die folgenden drei Restriktionen:

- Mindestens  $k$ :  $atLeast(k, \{b_0, b_1, \dots, b_{n-1}\})$
- Genau  $k$ :  $exactly(k, \{b_0, b_1, \dots, b_{n-1}\})$
- Höchstens  $k$ :  $atMost(k, \{b_0, b_1, \dots, b_{n-1}\})$

Anzahl-Constraints sind nicht Teil des SMTLib-Standards Version 2.6 [BFT17], sodass nur wenige Solver wie Z3 solche Kodierungen direkt unterstützen [Bjø+19]. Daher wird zur manuellen Kodierung oft die ITE-Kodierung verwendet [KGK19]. Dabei werden die einzelnen booleschen Unbekannten mithilfe eines If-Then-Else-Ausdrucks (kurz ITE) in numerische Ausdrücke umgewandelt, anschließend summiert und verglichen. Mit der Hilfsfunktion  $count(\{b_0, b_1, \dots, b_{n-1}\})$  lassen sich die Anzahl-Constraints einfach kodieren:

$$\text{count}(B) = \sum_{b \in B} \left( \begin{cases} 1 & , b \\ 0 & , \text{sonst} \end{cases} \right)$$

$$\text{atLeast}(k, B) = k \leq \text{count}(B)$$

$$\text{exactly}(k, B) = k = \text{count}(B)$$

$$\text{atMost}(k, B) = k \geq \text{count}(B)$$

Je nach gewählter Logik erfolgt die Summierung in der entsprechenden Sorte. Ist für das SMT-Problem beispielsweise die Logik QF\_LRA gewählt, dann summiert  $\text{count}(B)$  Terme der Sorte **Real**. Dabei entstehen für jedes Anzahl-Constraint offensichtlich  $n$  ITE-Ausdrücke,  $n - 1$  binäre Additionen und ein Vergleich.

Für bestimmte  $k$  wie beispielsweise  $k = 0$  oder  $k = 1$  können Anzahl-Constraints allerdings von spezialisierten Kodierungen profitieren. Einige davon sind folgende:

- $\text{atLeast}(0, B) = \text{True}$
- $\text{atLeast}(1, B) = \bigvee_{b \in B} b$
- $\text{atMost}(0, B) = \neg \bigwedge_{b \in B} b$
- $\text{atMost}(1, B) = \text{productEncoding}(B)$
- $\text{exactly}(0, B) = \neg \bigwedge_{b \in B} b$
- $\text{exactly}(1, B) = \text{atLeast}(1, B) \wedge \text{atMost}(1, B)$

Die Produkt-Kodierung (englisch *product-encoding*) ist eine aus SAT bekannte Kodierung für  $\text{atMost}(1, B)$  [Che10]. Dabei werden die  $n = |B|$  Unbekannten in eine Matrix  $M^{q \times p}$  mit  $p = \lceil \sqrt{n} \rceil$  und  $q = \lceil n/p \rceil$  gelegt. Das entspricht der Funktion  $f : B \rightarrow M$  mit  $f(b) = M_{j,i} \mapsto b$ . Zusätzlich wird für jeden Zeilenindex  $j \in \{0, \dots, q - 1\}$  eine Hilfsvariable  $v_j$  und jeden Spaltenindex  $i \in \{0, \dots, p - 1\}$  eine Hilfsvariable  $u_i$  angelegt, welche repräsentiert, ob in der jeweiligen Zeile  $j$  oder Spalte  $i$  mindestens ein  $b \in B$  wahr ist. Fasst man die Hilfsvariablen zu  $U = \{u_0, \dots, u_{p-1}\}$  und  $V = \{v_0, \dots, v_{q-1}\}$  zusammen, beschreibt folgende Gleichung, dass höchstens eine der  $n$  Unbekannten wahr ist:

$$\text{productEncoding}(B) =$$

$$\text{atMost}(1, U) \wedge \text{atMost}(1, V) \wedge \bigwedge_{0 \leq x \leq n-1, x=(i-1)(q-1)+j} ((\neg b_x \vee u_i) \wedge (\neg b_x \vee v_j))$$

Daraus folgen  $2\sqrt{n} + \mathcal{O}(\sqrt[4]{n})$  Hilfsvariablen und  $2n + 4\sqrt{n} + \mathcal{O}(\sqrt[4]{n})$  Klauseln. Das ist die nachweisbar minimale notwendige Anzahl an Klauseln für das At-Most-One-Constraint [Che10; KSV19].

Aus softwaretechnischer Sicht und wie später in Abschnitt 4.2.1 demonstriert, wäre eine Kodierung ohne Hilfsvariablen von Vorteil. Die Produkt-Kodierung kann

dazu einfach angepasst werden, indem alle Hilfsvariablen  $u_i \in U$  und  $v_j \in V$  folgend ersetzt werden:

$$u_i \mapsto \bigvee_{m_{j,i} \in M^{q \times p}, 0 \leq j \leq q-1} m_{j,i} \quad \text{und} \quad v_j \mapsto \bigvee_{m_{j,i} \in M^{q \times p}, 0 \leq i \leq p-1} m_{j,i}$$

Das eliminiert die Hilfsvariablen und behält die Klauselanzahl bei. Allerdings sind die entstehenden Klauseln um  $\mathcal{O}(\sqrt{n})$  größer.

### 3.4. Optimierung von SMT-Problemen

nuZ3 Paper hier sehr hilfreich [BPF15]

## 4. Haskell und Hasmtlib

In diesem Kapitel werden einige notwendige Grundlagen der funktionalen Programmiersprache Haskell erklärt. Danach wird der Kern der Funktionsweise der Haskell-Bibliothek `Hasmtlib` [Bru24a] beschrieben. Ferner werden einige geeignete Erweiterungen dieser diskutiert.

### 4.1. Funktionale Programmierung mit Haskell

Was ist ein ADT? Polymorphismus. Pattern-Matching. Was ist ein GADT und was kann er besser als ein ADT? Typ- und Kind-System anreißen - Phasentrennung. Typklassen. Typeoperators. Was ist eine pure Funktion? Wie kann man Nebenwirkungen dennoch modellieren - was ist eine Monade? Was ist Lazy?

### 4.2. Hasmtlib - Haskell-Bibliothek für SMTLib

Die Haskell-Bibliothek `Hasmtlib` [Bru24a] ist eine eingebettete domänenspezifische Sprache (englisch *embedded domain-specific language* - kurz *eDSL*) für den SMTLib-Standard Version 2.6 [BFT17] in Haskell. Eine eDSL ist eine, in einer Gastsprache eingebettete, anwendungsspezifische Sprache, welche auf Grundkonstrukten der Gastsprache aufbaut und somit den Implementierungsaufwand einer eDSL erheblich reduziert [SCK04].

Haskell wird aufgrund seines starken Typsystems oft als Gastsprache für solche eingebetteten Sprachen verwendet [Gil14]. Listing 4 zeigt beispielsweise, wie `Hasmtlib` SMTLib-Ausdrücke als abstrakten Syntaxbaum mit einem GADT repräsentiert.



```

1 data SMTSort = BoolSort | IntSort | RealSort | ...
2
3 data Expr (t :: SMTSort) where
4   Var      :: SMTVar t -> Expr t
5   Constant :: Value  t -> Expr t
6   Plus     :: Num (HaskellType t) => Expr t -> Expr t -> Expr t
7   ...
8   IDiv     :: Expr IntSort -> Expr IntSort -> Expr IntSort
9   Div      :: Expr RealSort -> Expr RealSort -> Expr RealSort
10  LTH      :: (Ord (HaskellType t), KnownSMTSort t) =>
11             Expr t -> Expr t -> Expr BoolSort
12  ...
13  Not      :: Boolean (HaskellType t) => Expr t -> Expr t
14  And      :: Boolean (HaskellType t) => Expr t -> Expr t -> Expr t
15  ...
16  Sqrt     :: Expr RealSort -> Expr RealSort

```

Listing 4: Repräsentation von SMTLib-Ausdrücken mithilfe eines GADTs in Hasmtlib [Bru24b]

Dabei ist der GADT `Expr t` polymorph in der `SMTSort`, sodass dessen Kind `Expr t :: SMTSort -> Type` ist. Da Werte des Datentyps `SMTSort` hier als Typen verwendet werden, nennt man `SMTSort` auch *Data-Kind* oder *Promoted-Type* [EW12]. Besonders relevant dafür ist das in 4.1 erklärte *Type-Refinement*. So weist beispielsweise ein Pattern-Match auf den Konstruktor `Sqrt` bereits zur Übersetzungszeit nach, dass `t ~ RealSort` gilt. Bei anderen Konstruktoren wie `And` kann die Belegung des `t` in `Expr t` mithilfe von Singletons ermittelt werden. Ein *Singleton* ist ein Laufzeit-Zeuge (Wert) für genau einen Übersetzungszeit-Typen [EW12]. Diese Abhängigkeit ist aufgrund Haskell's strenger Phasentrennung notwendig und ermöglicht eine *dependent*-ähnliche Kodierung wie in den Dependently-Typed Programmiersprachen Idris [BRA13] und Agda [Nor09].

Jeder Typ vom Kind `SMTSort` erhält also einen Singleton (Wert). Wie in Listing 5 zu sehen, stellt Hasmtlib den Singleton-Typen `SSMTSort` ebenfalls als GADT dar.

```

1 data SSMTSort (t :: SMTSort) where
2   SIntSort  :: SSMTSort IntSort
3   SRealSort :: SSMTSort RealSort
4   SBoolSort :: SSMTSort BoolSort
5   ...

```

Listing 5: Repräsentation des Singleton-Typs `SSMTSort` in Hasmtlib [Bru24c]

Die Verbindung zwischen Typ und Wert wird mit der in Listing 6 dargestellten Typklasse `KnownSMTSort` erreicht.

```

1 class KnownSMTSort (t :: SMTSort) where sortSing :: SSMTSort t
2 instance KnownSMTSort IntSort      where sortSing = SIntSort
3 instance KnownSMTSort RealSort     where sortSing = SRealSort
4 instance KnownSMTSort BoolSort     where sortSing = SBoolSort
5 ...

```

Listing 6: Typklasse zur Verbindung von Typ SMTSort zu Wert SSMTSort in Hasmtlib [Bru24c]

Mit den genannten Mechanismen garantiert **Hasmtlib** die ausschließliche Erzeugung von sauber typisierten (englisch *well-typed*) Ausdrücken [Pey+06]. Diese werden anschließend von der Bibliothek in SMTLib-Syntax übersetzt und an einen SMT-Solver gegeben. Dessen Antwort parst und übersetzt **Hasmtlib** dann zurück in die entsprechenden Haskell-Typen. Folgend ein kleines Programm zur Erläuterung:

```

1 module Example.Foo where
2
3 import Language.Hasmtlib
4 import Control.Monad (replicateM)
5
6 data Bar a = Bar
7   { length :: a
8     , height :: a }
9   deriving (Generic)
10 instance Variable a => Variable (Bar a)
11
12 barLength :: [Bar (Expr RealSort)] -> Expr BoolSort
13 barLength = all (\(b1,b2) -> ...) . binom
14   where
15     binom = ...
16
17 main :: IO ()
18 main = do
19   result <- solveWith @SMT (solver cvc5) $ do
20     setLogic "QF_LRA"
21     -- create variables
22     bars <- replicateM 10 variable
23     -- constrain variables
24     assert $ barLength bars
25     -- find model for variables
26     return foo
27   ...

```

Listing 7: Beispielhafte Verwendung von Hasmtlib

Das Beispiel in Listing 7 demonstriert mit der Funktion `barLength` außerdem die pure Formelkonstruktion in **Hasmtlib**. Lediglich Variablenerzeugung (Zeile 20)

und Formel-Assertion (Zeile 22) sind monadisch. Dadurch lassen sich viele Funktionalitäten der Gastsprache wiederverwenden, zum Beispiel die Funktion `sum` aus `Data.Foldable`: `sum :: (Foldable t, Num a) => t a -> a`.

#### 4.2.1. Ergänzung um Anzahl-Constraints

Hasmtlib unterstützte bis zum Zeitpunkt dieser Arbeit keine Anzahl-Constraints. Daher werden diese im Rahmen dieser Arbeit der Bibliothek hinzugefügt. Die Implementierung erfolgt analog zu den in 3.3 präsentierten Kodierungen. Folgend ein Ausschnitt für das At-Most-Constraint:

```

1  count :: forall t f. (Functor f, Foldable f, Num (Expr t))
2    => f (Expr BoolSort) -> Expr t
3  count = sum . fmap (\b -> ite b 1 0)
4
5  atMost :: forall t f. (Functor f, Foldable f, KnownSMTSort t
6    , Num (HaskellType t), Ord (HaskellType t))
7    => Expr t -> f (Expr BoolSort) -> Expr BoolSort
8  atMost 0 = nand
9  atMost 1 = amoSqrt
10 atMost k = (<=? k) . count
11
12 -- Pure product-encoding
13 amoSqrt :: (Foldable f, Boolean b) => f b -> b
14 amoSqrt xs
15   | length xs < 10 = amoQuad $ toList xs -- pair-wise encoding
16   | otherwise =
17     let n = toInteger $ length xs
18         p = ceiling $ sqrt $ fromInteger n
19         rows = splitEvery (fromInteger p) $ toList xs
20         columns = transpose rows
21         vs = or <$> rows
22         us = or <$> columns
23     in amoSqrt vs && amoSqrt us &&
24       and (
25         imap
26           (\j r -> and $
27             imap (\i x -> (x ==> us !! i) && (x ==> vs !! j)) r)
28           rows
29       )
30   where
31     splitEvery n = takeWhile (not . null) . map (take n) . iterate (drop n)

```

Listing 8: ITE-Kodierung von Anzahl-Constraints in Hasmtlib [Bru24d]

#### 4.2.2. Ergänzung um Incremental Refinement

## 5. Implementierung

In diesem Kapitel wird gezeigt, wie das Problem der Pressenplanung mit Haskell und `HasmtLib` in `SMTLib` kodiert werden kann. Dazu werden zunächst die notwendigen Datentypen und die Sorten-Polymorphie der Kodierung erklärt, bevor einige konkrete Repräsentationen der Constraints aus Kapitel 2 Abschnitt 2.2 mithilfe von Quellcode-Ausschnitten dargestellt werden. Danach werden Ein- und Ausgabe kurz beschrieben. Anschließend wird dargelegt, wie die Implementierung mit bestimmten Parametern konfiguriert werden kann und wie diese die Laufzeit des Solvers beeinflussen. Am Ende wird eine Möglichkeit zur Parallelisierung des Programms demonstriert.

### 5.1. Datentypen und Variablen

Analog zur Problemdomäne aus Kapitel 2 sind hier Bauteile (Components), Schichten (Layers) und Pressen (Presses) die Entitäten von Bedeutung. Deren Repräsentation erfolgt jeweils durch Record-Datenstrukturen. Bei Pressen wird zusätzlich in statische und symbolische Pressen unterschieden. Statische Pressen (`StaticPress`) beschreiben die konstanten Eigenschaften von Pressen wie beispielsweise deren maximale Höhe. Symbolische Pressen (`Press`) haben die Konstanten einer statischen Presse und beschreiben weiter symbolische Eigenschaften wie die geplante Länge in einem Pressenplan. Diese Unterscheidung ist für die Trennung von Eingabe (Einlesen der Pressenspezifikationen) und Verarbeitung der Pressen im Constraint-System sinnvoll. Folgend deren Haskell-Repräsentation:

```

1  data Component a = Component
2    { _cId          :: Integer
3    , _compLength  :: a
4    , _compHeight  :: a
5    } deriving stock (Show, Eq, Ord, Functor, Foldable, Traversable, Generic)
6    deriving anyclass (Codec)
7  $(makeLenses ''Component)
8
9  data Layer a b = Layer
10   { _lId          :: Integer
11   , _layerLength  :: a
12   , _layerHeight  :: a
13   , _layerEmpty   :: b
14   , _waste        :: a
15   } deriving stock (Show, Eq, Ord, Generic)
16   deriving anyclass (Codec)
17  $(makeLenses ''Layer)
18
19  data StaticPress a = StaticPress
20   { _pId          :: Integer
21   , _pressName     :: String
22   , _pressMinLength :: a
23   , _pressMaxLength :: a
24   , _pressMinHeight :: a
25   , _pressMaxHeight :: a
26   } deriving stock (Show, Eq, Ord, Functor, Foldable, Traversable, Generic)
27   deriving anyclass (Codec)
28  $(makeLenses ''StaticPress)
29
30  data Press a b = Press
31   { _press        :: StaticPress a
32   , _pressLength  :: a
33   , _pressHeight  :: a
34   , _pressLayers  :: [Layer a b]
35   } deriving stock (Show, Eq, Ord, Generic)
36   deriving anyclass (Codec)
37  $(makeLenses ''Press)

```

Listing 9: Haskell-Datentypen der Domäne Pressenplanung

Der Ausdruck `$(makeLenses ''DATATYPE)` taucht dabei mehrmals auf. Dieser erzeugt *Lenses* für die Datentypen. Da im Weiteren Lenses häufig in der Implementierung verwendet werden, ist der Leser angeraten, folgende Quelle bei Fragen zu Lenses zu konsultieren: [Kme24].

Ferner ist auffällig, dass alle Typen polymorph sind, sodass `Layer a b` beispielsweise zwei Typen `a` und `b` nimmt. Das `a` repräsentiert einen Zahlentypen und das `b` einen Bool-Typen. Das vereinfacht später die Kodierung und Dekodierung der symbolischen Werte. So könnte die symbolische Schicht mit `type SLayer t = Layer`

(Expr t) (Expr BoolSort) und die konkrete Schicht mit `type CLayer t = Layer (HaskellType t) Bool` dargestellt werden.

Mit diesen Grundtypen kann nun der Typ `Pressenplan` definiert werden:

```

1 data PressPlan t = PressPlan
2   { _components :: [Component (Expr t)]
3   , _layers      :: [Layer (Expr t) (Expr BoolSort)]
4   , _presses     :: [Press (Expr t) (Expr BoolSort)]
5   , _inLayer     :: Relation Integer Integer
6   } deriving stock (Generic)
7 $(makeLenses ''PressPlan)

```

Listing 10: Haskell-Datentyp des Pressenplans

Dieser enthält alle symbolischen Komponenten, Schichten, Pressen und auch die Relation *InLayer* aus Abschnitt 2.2. Als Index für diese wird das Tupel (cId, lId) aus Komponenten-ID und Schicht-ID verwendet.

Die Erzeugung eines symbolischen Pressenplans ist monadisch (`setupPressPlan :: MonadSMT s m => ... -> m (PressPlan t)`), da dort alle notwendigen Variablen angelegt werden. Identisch zur Variablenmenge  $\mathbb{X}$  in Abschnitt 2.2 sind das:

- `_pressLength` und `_pressHeight` für jede Presse,
- `_layerLength`, `_layerHeight`, `_waste` und `_layerEmpty` für jede Schicht und
- ein symbolischer Wahrheitswert für jedes mögliche Element der Relation *InLayer*

Bei weiterer Betrachtung des Typs `Pressenplan` fällt auf, dass Komponenten, Schichten und Pressen im Typ Liste vorliegen. Der Listentyp ist der von Haskell, also konkret in Haskell und nicht symbolisch im SMT-Problem. Das erklärt die Nutzung der Liste für die Komponenten, da diese eine Eingabe sind, welche konkret darstellbar ist. Aber wie ist der Listentyp für Schichten und Pressen zu erklären?

Das Modell in Abschnitt 2.2 all-quantifiziert Schichten und Pressen. Das wäre analog auch in SMT möglich. Im Allgemeinen werden solche Probleme aber grundinstanziiert, um die Verwendung ineffizient (oder gar nicht) entscheidbarer Quantoren-Logiken zu vermeiden [GBT07; GM09]. Für jede Presse wird dabei im Voraus die höchstmögliche Anzahl an Schichten basierend auf der Eingabe an Komponenten ermittelt, genauer in 5.2. Da Schichten leer sein können haben sie das Attribut `_layerEmpty :: b` (Listing 9, Zeile 13). Identisch könnten Pressenpläne leere Pressen enthalten. Weiter könnte eine obere Schranke für die Anzahl an Pressen basierend auf der Eingabe an Komponenten ermittelt werden. Allerdings bedeutet jede weitere Presse weitere Unbekannte und weitere Constraints, sodass die Komplexität des Problems erhöht wird, was letztendlich zu einer ineffizienteren Lösung des Solvers führt. Daher wird jeder Auftrag mehrmals kodiert - mit jeweils unterschiedlich fester Anzahl an Pressen. Diese werden dann als separate Probleminstanzen auf verschiedene Kerne aufgeteilt, mehr dazu in Abschnitt 5.6.

## 5.2. Vorverarbeitung

## 5.3. Sorten-polymorphe Kodierung

Wie in Kapitel 3 gezeigt bietet SMTLib verschiedene Logiken zur Kodierung von SMT-Problemen an. Nicht immer ist dabei vorher klar, in welcher Logik ein Problem am effizientesten für Solver kodiert werden kann. Das gilt auch für das Pressenplanungsproblem mit den Kandidaten `QF_LIA`, `QF_LRA` und `QF_BV`. Hasmtlibs Ausdrucks-Typ `Expr t` kann dafür genutzt werden, um das Problem polymorph in der Sorte zu implementieren. Betrachten wir folgende Funktion, die für einen Pressenplan und dessen Konfiguration alle Constraints für jenen erstellt:

```
1 problem :: (Orderable (Expr t), Num (Expr t))
2     => PressPlan t
3     -> Config (Expr t)
4     -> [Expr BoolSort]
5 problem pp cfg = ...
```

Listing 11: Sorten-Polymorphie bei der Problemdeklaration

Der Aufrufer wählt die Sorte der Kodierung dann per Typ-Applikation und setzt die Logik mithilfe der Singletons:

```
1 runProblem = forall t. (KnownSMTSort t, MonadSMT s m) => ... -> m ()
2 runProblem ... = do
3     ...
4     setLogic $ case sortSing @t of
5         SIntSort      -> "QF_LIA"
6         SRealSort     -> "QF_LRA"
7         SBvSort _ _   -> "QF_BV"
8         _             -> "ALL"
9     ...
10    forM_ (problem @t _ _) assert
```

Listing 12: Auswahl der Logik nach Sorte

Das ermöglicht die Wahl der Kodierung für den Aufrufer der Schnittstelle und verhindert somit Redundanz in der Implementierung.

## 5.4. Kodierung der Constraints

## 5.5. IO

Ein- und Ausgabe sind wenig interessant und werden der Vollständigkeit wegen hier nur kurz erwähnt. Germanedge's bisherige heuristische Lösung spezifiziert Ein- und Ausgabeformat als CSV. Diese Spezifikation wird hier ebenfalls implementiert.

Dazu wird die Haskell-Bibliothek `cassava` [Tib24] verwendet. Folgend das Lesen eines Auftrages und das Schreiben der Lösungen für diesen:

```
1 data ComponentIn = ComponentIn
2   { length    :: !Integer
3   , height    :: !Integer
4   , count     :: !Integer
5   } deriving stock (Generic)
6   deriving anyclass (FromRecord, FromNamedRecord, DefaultOrdered)
7
8 data ComponentOut = ComponentOut
9   { press      :: !Integer
10  , layer       :: !Integer
11  , position    :: !Integer
12  , length      :: !Integer
13  , height      :: !Integer
14  } deriving stock (Generic)
15  deriving anyclass (ToRecord, ToNamedRecord, DefaultOrdered)
16
17 -- Schedule is the representation for the decoded Solution
18 fromSchedule :: Schedule -> [ComponentOut]
19 fromSchedule = ...
20
21 main :: IO ()
22 main = do
23   csvOrder <- ByteString.Lazy.readFile "input/order.csv"
24   let order = case Csv.decode @ComponentIn Csv.HasHeader csvOrder of
25               Left err -> error err
26               Right v  -> toList v
27   ...
28   schedules <- disperseProblem order ...
29   forM_ schedules
30     (\s -> ByteString.Lazy.writeFile (scheduleName s <> ".csv") $
31       Csv.encodeDefaultOrderedByName $ fromSchedule s)
```

Listing 13: CSV-Verarbeitung mithilfe der Haskell-Bibliothek `cassava`

## 5.6. Parallelisierung

## 5.7. Modifikation und Konfiguration der Kodierung

Welche vielversprechenden Anpassungen habe ich noch probiert, die aber keine signifikanten Verbesserungen der Solver-Laufzeit gegeben haben?



## **6. Auswertung und Laufzeitmessung**

Was haben wir für Testfälle? Woher kommen die Testfälle? Wie wurden die Testfälle ausgewählt? Sind sie repräsentativ für reale Probleme? Vorgehen Laufzeitmessung erwähnen. Maschinen-Specs nennen.

### **6.1. Vergleich mit realistischem Timeout**

Time-Limit 5 Minuten, Vergleich Heuristik Germanedge vs SMT. Nur ein Modell oder beide? - Beide. Auch OMT? - Ja. Auswertung der Ergebnisse.

### **6.2. Vergleich mit liberalem Timeout**

Wer findet bessere Lösung, wenn Timeout sehr liberal (24h)? Hier werden Hardware-Specs ein Problem. Das muss auf Pool-PCs der HTWK gemacht werden. Darf da ein Binary von der Heuristik laufen? Auswertung der Ergebnisse.

## **7. Zusammenfassung**

## Literatur

- [Ans+11] Carlos Ansótegui u. a. „Satisfiability Modulo Theories: An Efficient Approach for the Resource-Constrained Project Scheduling Problem“. In: Jan. 2011, S. 2–9. ISBN: 9781577355434.
- [Bar+22] Haniel Barbosa u. a. „cvc5: A Versatile and Industrial-Strength SMT Solver“. In: *Tools and Algorithms for the Construction and Analysis of Systems*. Hrsg. von Dana Fisman und Grigore Rosu. Cham: Springer International Publishing, 2022, S. 415–442. ISBN: 978-3-030-99524-9.
- [BFT17] Clark Barrett, Parcal Fontaine und Cesare Tinelli. *The SMT-LIB Standard Version 2.6*. Techn. Ber. Abgerufen am: 21.08.2024. Department of Computer Science, The University of Iowa, 2017. URL: <https://smt-lib.org/papers/smt-lib-reference-v2.6-r2017-07-18.pdf>.
- [Bjø+19] Nikolaj Bjørner u. a. „Programming Z3“. In: Springer Nature, Apr. 2019, S. 148–201.
- [BMS05] Clark Barrett, Leonardo de Moura und Aaron Stump. „SMT-COMP: Satisfiability Modulo Theories Competition“. In: *Computer Aided Verification*. Hrsg. von Kousha Etessami und Sriram K. Rajamani. Berlin, Heidelberg: Springer Berlin Heidelberg, 2005, S. 20–23. ISBN: 978-3-540-31686-2.
- [BPF15] Nikolaj S. Bjørner, Anh-Dung Phan und Lars Fleckenstein. „ $\nu$ Z - An Optimizing SMT Solver“. In: *International Conference on Tools and Algorithms for Construction and Analysis of Systems*. 2015. URL: <https://api.semanticscholar.org/CorpusID:17542710>.
- [BRA13] EDWIN BRADY. „Idris, a general-purpose dependently typed programming language: Design and implementation“. In: *Journal of Functional Programming* 23 (Sep. 2013). DOI: 10.1017/S095679681300018X.
- [Bru+10] Roberto Bruttomesso u. a. „The OpenSMT Solver“. In: *Tools and Algorithms for the Construction and Analysis of Systems*. Hrsg. von Javier Esparza und Rupak Majumdar. Berlin, Heidelberg: Springer Berlin Heidelberg, 2010, S. 150–153. ISBN: 978-3-642-12002-2.
- [Bru24a] Julian Bruder. *hasmtlib: A monad for interfacing with external SMT solvers*. Version 2.3.2. Abgerufen am: 20.08.2024. 2024. URL: <https://hackage.haskell.org/package/hasmtlib>.
- [Bru24b] Julian Bruder. *hasmtlib: A monad for interfacing with external SMT solvers*. Version 2.4.0. Abgerufen am: 25.08.2024. 2024. URL: <https://hackage.haskell.org/package/hasmtlib-2.4.0/docs/src/Language.Hasmtlib.Type.Expr.html#Expr>.

- [Bru24c] Julian Bruder. *hasmtlib: A monad for interfacing with external SMT solvers*. Version 2.4.0. Abgerufen am: 25.08.2024. 2024. URL: <https://hackage.haskell.org/package/hasmtlib-2.4.0/docs/src/Language.Hasmtlib.Type.SMTSort.html#SSMTSort>.
- [Bru24d] Julian Bruder. *hasmtlib: A monad for interfacing with external SMT solvers*. Version 2.5.0. Abgerufen am: 26.08.2024. 2024. URL: <https://hackage.haskell.org/package/hasmtlib-2.5.0/docs/src/Language.Hasmtlib.Counting.html#count%27>.
- [BT18] Clark Barrett und Cesare Tinelli. „Satisfiability Modulo Theories“. In: *Handbook of Model Checking*. Hrsg. von Edmund M. Clarke u. a. Cham: Springer International Publishing, 2018, S. 305–343. ISBN: 978-3-319-10575-8. DOI: 10.1007/978-3-319-10575-8\_11. URL: [https://doi.org/10.1007/978-3-319-10575-8\\_11](https://doi.org/10.1007/978-3-319-10575-8_11).
- [Che10] Jingchao Chen. „A New SAT Encoding of the At-Most-One Constraint“. In: 2010. URL: <https://api.semanticscholar.org/CorpusID:15322159>.
- [DB08] Leonardo De Moura und Nikolaj Bjørner. „Z3: an efficient SMT solver“. In: *Proceedings of the Theory and Practice of Software, 14th International Conference on Tools and Algorithms for the Construction and Analysis of Systems*. TACAS’08/ETAPS’08. Budapest, Hungary: Springer-Verlag, 2008, S. 337–340. ISBN: 3540787992.
- [Dut14] Bruno Dutertre. „Yices 2.2“. In: *Computer Aided Verification*. Hrsg. von Armin Biere und Roderick Bloem. Cham: Springer International Publishing, 2014, S. 737–744. ISBN: 978-3-319-08867-9.
- [EW12] Richard A. Eisenberg und Stephanie Weirich. „Dependently typed programming with singletons“. In: *Proceedings of the 2012 Haskell Symposium*. Haskell ’12. Copenhagen, Denmark: Association for Computing Machinery, 2012, S. 117–130. ISBN: 9781450315746. DOI: 10.1145/2364506.2364522. URL: <https://doi.org/10.1145/2364506.2364522>.
- [GBT07] Yeting Ge, Clark Barrett und Cesare Tinelli. „Solving Quantified Verification Conditions Using Satisfiability Modulo Theories“. In: Apr. 2007, S. 167–182. ISBN: 978-3-540-73594-6. DOI: 10.1007/978-3-540-73595-3\_12.
- [Gil14] Andy Gill. „Domain-specific Languages and Code Synthesis Using Haskell: Looking at embedded DSLs“. In: *Queue* 12.4 (Apr. 2014), S. 30–43. ISSN: 1542-7730. DOI: 10.1145/2611429.2617811. URL: <https://doi.org/10.1145/2611429.2617811>.
- [GM09] Yeting Ge und Leonardo de Moura. „Complete Instantiation for Quantified Formulas in Satisfiability Modulo Theories“. In: *Computer Aided Verification*. Hrsg. von Ahmed Bouajjani und Oded Maler. Berlin, Heidelberg: Springer Berlin Heidelberg, 2009, S. 306–320. ISBN: 978-3-642-02658-4.
- [Hud+07] Paul Hudak u. a. „A history of Haskell: Being lazy with class“. In: Juni 2007, S. 1–55. DOI: 10.1145/1238844.1238856.

- [KGK19] Gergely Kovásznai, Krisztián Gajdár und Laura Kovács. „Portfolio SAT and SMT Solving of Cardinality Constraints in Sensor Network Optimization“. In: *2019 21st International Symposium on Symbolic and Numeric Algorithms for Scientific Computing (SYNASC)*. 2019, S. 85–91. DOI: 10.1109/SYNASC49474.2019.00021.
- [Kme24] Edward A. Kmett. *lens: Lenses, Folds and Traversals*. Version 5.3.2. Abgerufen am: 16.09.2024. 2024. URL: <https://hackage.haskell.org/package/lens>.
- [KSV19] Petr Kučera, Petr Savický und Vojtěch Vorel. „A lower bound on CNF encodings of the at-most-one constraint“. In: *Theoretical Computer Science* 762 (2019), S. 51–73. ISSN: 0304-3975. DOI: <https://doi.org/10.1016/j.tcs.2018.09.003>. URL: <https://www.sciencedirect.com/science/article/pii/S0304397518305681>.
- [LLC21] Daeyeal Lee, Bill Lin und Chung-Kuan Cheng. „SMT-Based Contention-Free Task Mapping and Scheduling on 2D/3D SMART NoC with Mixed Dimension-Order Routing“. In: *ACM Trans. Archit. Code Optim.* 19.1 (Dez. 2021). ISSN: 1544-3566. DOI: 10.1145/3487018. URL: <https://doi.org/10.1145/3487018>.
- [Nor09] Ulf Norell. „Dependently Typed Programming in Agda“. In: Jan. 2009, S. 1–2. ISBN: 978-3-642-04651-3. DOI: 10.1007/978-3-642-04652-0\_5.
- [Pey+06] Simon Peyton Jones u. a. „Simple Unification-based Type Inference for GADTs“. In: *International Conference on Functional Programming (ICFP)*. ACM, 2006. DOI: 10.1145/1159803.1159811.
- [SCK04] Sean Seefried, Manuel Chakravarty und Gabriele Keller. „Optimising Embedded DSLs Using Template Haskell“. In: *Generative Programming and Component Engineering*. Hrsg. von Gabor Karsai und Eelco Visser. Berlin, Heidelberg: Springer Berlin Heidelberg, 2004, S. 186–205. ISBN: 978-3-540-30175-2.
- [Tib24] Johan Tibell. *cassava: A CSV parsing and encoding library*. Version 0.5.3.2. Abgerufen am: 16.09.2024. 2024. URL: <https://hackage.haskell.org/package/cassava>.
- [VAB18] Saurabh Vaidya, Prashant Ambad und Santosh Bhosle. „Industry 4.0 – A Glimpse“. In: *Procedia Manufacturing* 20 (2018). 2nd International Conference on Materials, Manufacturing and Design Engineering (iCMMD2017), 11-12 December 2017, MIT Aurangabad, Maharashtra, INDIA, S. 233–238. ISSN: 2351-9789. DOI: <https://doi.org/10.1016/j.promfg.2018.02.034>. URL: <https://www.sciencedirect.com/science/article/pii/S2351978918300672>.

## **Appendix**

**A. Beispiel 1**

**⋮**

**B. Beispiel n**

## **Danksagung**

Zuallererst möchte ich mich bei Herrn Prof. Dr. Waldmann für das im Wintersemester 2023/2024, in der Vorlesung Constraint-Programmierung geweckte Interesse an jener bedanken. Vielen Dank auch für Ihre zahlreichen Hinweise und Vorschläge während dieser Bachelorarbeit. Weiter möchte ich meinen Dank der Germanedge Solutions GmbH und dort besonders Christoph Schumacher aussprechen, die diese Bachelorarbeit ermöglicht haben. Abschließend auch ein großes Dankeschön an Dr. Sebastian Möbius, welcher mir als Firmenbetreuer jederzeit alle Fragen bestens beantwortet hat.

## **Eidesstattliche Erklärung**

Hiermit versichere ich, dass ich die vorliegende Arbeit selbstständig ohne unzulässige Hilfe Dritter und ohne Benutzung anderer als der angegebenen Quellen und Hilfsmittel verfasst habe. Die den verwendeten Quellen und Hilfsmitteln wörtlich oder sinngemäß entnommenen Stellen sind als solche kenntlich gemacht. Die Arbeit wurde bisher keiner anderen Prüfungsbehörde vorgelegt und auch noch nicht veröffentlicht. Mir sind die strafrechtlichen Konsequenzen einer falschen eidesstattlichen Erklärung bekannt.

Leipzig, den 16. September 2024